Behavioural Fault Modelling using VHDL-AMS and Slow Transient Analysis with hAMSter Simulator to Speed-up Analogue Fault Simulation

Yavuz Kılıç (Member IEEE) and Mark Zwoliński
 (Senior Member IEEE)

Philips Semiconductors, Southampton, UK

E-mail: Yavuz.Kilic@philips.com
Keywords: macromodelling, behavioural modelling, VHDL-AMS, fault simulation
Abstract


Two of techniques to speed-up analogue fault simulation are: fault dropping/collapsing, in which faults that have similar circuit responses compared with the fault-free circuit response and/or with another faulty circuit response are considered equivalent; and behavioural/macro modelling, whereby parts of the circuit are modelled at a more abstract level, therefore reducing the complexity and the simulation time. This paper discusses behavioural modelling as a way to speed-up fault simulation for analogue circuits. In this paper, a new behavioural fault model is developed in VHDL-AMS for a CMOS operational amplifier circuit using slow transient analysis. Simulation results confirm up to 373 times speed-up in terms of CPU time.

1 . Introduction

One way to speed-up fault simulation for analogue and mixed-signal circuits is to use behavioural/macro models, where parts of the circuit are modelled at a more abstract level, reducing the complexity and hence the simulation time. Characterizing behavioural fault models requires low-level simulations and is therefore not applicable in every case. There are three situations in which behavioural fault modelling might be of benefit, however. First, if circuit blocks are reused, low-level fault simulations would not have to be repeated. Second, at the system design stage, information about possible faults and how their effects might be propagated can be used to insert optimal test structures. Third, actual faults can be modelled at a low level (analogue or digital), while neighbouring circuit blocks can be modelled behaviourally, again reducing the overall simulation time.

A behavioural model describes a circuit block in terms of mathematical equations modelling the functionality of the block, for example, in terms of the input-output relationship.  Behavioural modelling has been used for speeding up analogue simulation in general [1] and analogue fault simulation in particular [2]-[8]. In [1], analogue circuits were modelled behaviourally in the C programming language. Broyden’s method was used to formulate and solve the model equations in a custom simulator. Broyden’s method was originally proposed in [9] as an algorithm for the solution of systems of nonlinear equations, i.e. of the derivatives of a set of functions. The main drawback with the work described in [9] is that since the technique does not require derivatives it cannot be used for small-signal analysis. 


In [3], Chang et al presented a behavioural fault model derived from a macromodel of a CMOS operational amplifier from the IEEE Mixed-Signal Benchmark Suite

[22] (Figure 1). The behavioural model was implemented using an analogue HDL, MAST, in [7]. A VHDL-AMS model of the behavioural model was proposed in [6] and [8].


In this paper, the VHDL-AMS model proposed in [6] and [8] will be extended by using slow transient analysis in conjunction with the hAMSter simulator [10] to derive some conclusions for different fault types for the CMOS opamp shown in Figure 1. 


Figure 1. The 2-stage CMOS Miller opamp used in [3] for behavioural fault modelling.


The rest of the paper is organised as follows. In Section 2 behavioural model for inverting operational amplifier circuit is discussed and a VHDL-AMS model is developed. In Section 3 the simulation results are presented. Finally in Section 4 some conclusions are drawn.

2 . Behavioural Model

The input-output relationship for the CMOS opamp was derived in [3] as

Table I. The values of the parameters m and k for different fault types for the closed-loop inverting opamp behavioural model.

	Fault types
	Parameters

	
	m
	k [V]

	Type I
	-1.02
	2.15

	Type II
	0
	0.011

	Type III
	0 if vin > 0V

-2 if vin < 0
	0

	Type IV
	-1 if vin > ~1.2V and vin < ~ -1.2V

0 if ~ -1.2V < vin < ~ 1.2V
	Vdd/2 if vin > ~ 1.V

Vss/2 if vin < ~ -1.2V

0 if ~ -1.2V < vin < ~ 1.2V

Figure 2. Behavioural level DC-offset fault model for the inverting opamp.

where ACL is the closed-loop gain of the opamp, and m and k are parameters to model four fault types for the opamp. The values of m and k were derived by carrying out transistor level simulations of the four fault types and are given in Table I.


By considering the input-output relationship, given above, the faulty opamp macromodel given in [3] can be simplified to that shown in Figure 2. All the fault effects and non-ideal effects are approximated to Fos = mVin + k, which is applied to the inverting input of the opamp.


A VHDL-AMS implementation of the behavioural model given in (1) is given in Figure 3 and Figure 4 [11]. rin represents the input resistance of the opamp in Figure 3 where it is only used for the third equation in Figure 4. The third equation is needed for completeness as there are three quantities declared in the architecture declaration shown in Figure 4. Note that the architecture declaration given in Figure 4 also covers the supply voltage limiting effect at the output of the opamp.

--behavioural opamp

library disciplines;                        

library ieee;

use disciplines.electromagnetic_system.all;

use ieee.math_real.all;

--entity

entity op_behav is                          

   generic ( m : real := 0.0; --fault-free value

           k : real := 0.0; --fault-free value

           Acl: real:= -1.0; --closed-loop gain

           rin : real := 100.0e6);

       port (terminal in_node, out_node : electrical);

end;
Figure 3. The VHDL-AMS entity implementation of the behavioural fault model.

--architecture 

library disciplines;

library ieee;

use disciplines.electromagnetic_system.all;

use ieee.math_real.all; 

architecture behav of op_behav is
   quantity vout across iout through out_node;

   quantity vin across iin through in_node;

   quantity Fos : real;

   -- supply voltage limit

   constant v_limit : real := 2.5;

begin 

procedural is

  variable vout_calc : real;

  begin

   Fos := m*vin + k;   

   vout_calc := Acl * (vin + Fos); 

   iin  := (vin - Fos) / rin;

   if (vout_calc > v_limit) then vout := 2.5;

   elsif (vout_calc < -v_limit) then vout := -2.5;

   else vout := vout_calc;

   end if; 

end procedural; 

end;
Figure 4. The VHDL-AMS architecture implementation of the behavioural fault model.


In order to simulate the VHDL-AMS model shown in Figure 3 and Figure 4, one also needs VHDL-AMS models for a resistor, a voltage source and a testbench, which are shown in Figure 5, Figure 6, and Figure 7, respectively.

--resistor

library disciplines;

use disciplines.electromagnetic_system.all;

entity resistor is
  generic (rnom : real := 0.0); 

  port(terminal p,m  : electrical); --interface ports.

end resistor;

architecture behav of resistor is
    quantity r_e across r_i through p to m;

begin

    r_i == r_e/rnom; 

end behav;
Figure 5. A VHDL-AMS model of a resistor.
-- voltage source

library disciplines;

use disciplines.electromagnetic_system.all;

--entity declaration.

entity v_source is
   generic (dc_value : real :=-2.50);


port(terminal p,m: electrical);--interface ports.

end v_source;                                                                

--architecture declaration.

architecture behav of v_source is
    quantity v_in across i_out through p to m;

begin

  v_in==dc_value*now; -- slow transient 

end architecture behav;
Figure 6. A VHDL-AMS model of a voltage source.


Note that input voltage source in architecture declaration shown in Figure 6 is realised using a predefined VHDL-AMS function, now, which returns the value of the current time at each step as simulation proceeds. This is done in order to simulate the DC-sweep analysis, which is not defined in VHDL-AMS (unlike in SPICE-like simulators).  This technique is called slow transient simulation.

--testbench

library disciplines;

library ieee;

use disciplines.electromagnetic_system.all;

use ieee.math_real.all;

entity ex_op_behav is end;

architecture testbench of ex_op_behav is

  terminal in_node, out_node, vsrc : electrical;

begin                                  

  op_behav_uut : entity op_behav (behav)   

    generic map ( m => 0.0,k => 0.0, acl => -1.0)

    port map ( in_node  => in_node, 

               out_node => out_node);

  vin_dc : entity v_source (behav) 

  generic map ( dc_value => 5.0)

  port map ( p => vsrc, m => electrical_ground);

  rsrc : entity resistor (behav)

    generic map (rnom => 100.0)

    port map (p => vsrc, m => in_node);

end;
Figure 7. A VHDL-AMS testbench used with the hAMSter simulator to simulate the behavioural model shown in Figure 3 and Figure 4.

3 . Simulation Results


The slow transient simulation results using the hAMSter [10] simulator and the behavioural closed-loop VHDL-AMS model of the inverting opamp (the fault-free case) with the necessary component and voltage source models and the testbench shown in Figure 3-Figure 7 are shown in Figure 8 and Figure 9 for both the positive and the negative values of the input voltage source. 

Figure 8. Slow-transient simulation of the VHDL-AMS model with hAMSter for positive input source values.

Figure 9. Slow-transient simulation of the VHDL-AMS model with hAMSter for negative input source values.


Note that the X-axis in Figure 8 and Figure 9 represents the time in seconds, where Y-axis represents vout, vin, and Fos in Volts. (Unless otherwise stated, for the rest of the paper it will be assumed that X-axis will represent time in seconds for the simulation results obtained using hAMSter).


Using the values for the parameters m and k from Table I yields the simulation results shown in Figure 10-Figure 15 and Figure 18-Figure 19.

Figure 10. The VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type I faults for the positive values of vin.

Figure 11. VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type I faults for negative values of vin.

Figure 12. VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type II faults for positive values of vin.

Figure 13. VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type II faults for negative values of vin.

Figure 14. VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type III faults for positive values of vin.

Figure 15. VHDL-AMS behavioural closed-loop inverting opamp model slow-transient simulation using hAMSter for Type III faults for negative values of vin.


Note that the output response of the opamp, vout, found for the positive values of vin and the negative values of vin for Type I faults (Figure 10 and Figure 11) are the same (“nearly stuck-at” –2.14 V), as the expected results [3]. 


For Type III faults Fos is determined using the if-then construct in the VHDL-AMS model shown in Figure 16. Similarly, for Type IV faults Fos is determined using the if-then construct in the VHDL-AMS model shown in Figure 17.

if (vin < 0.0) then
   Fos := -2.0*vin;

else        

   Fos := 0.0;

end if;  

Figure 16. if-then construct implemented in the VHDL-AMS model for Type III faults.

if (vin > 1.2) then
   Fos := -vin + 1.25;

elsif (vin < -1.2) then
   Fos := -vin - 1.25;

else        

   Fos := 0.0;

end if; 

Figure 17. if-then construct implemented in the VHDL-AMS model for Type IV faults.

Figure 18. VHDL-AMS behavioural closed-loop non-inverting opamp model slow-transient simulation using hAMSter for Type IV faults for positive values of vin.

Figure 19. VHDL-AMS behavioural closed-loop non-inverting opamp model slow-transient simulation using hAMSter for Type IV faults for negative values of vin.


As DC-sweep analysis cannot be performed for VHDL-AMS, the transient simulation results for different fault types using VHDL-AMS behavioural models and hAMSter simulator were compared with the transient simulation results that are obtained using transistor level models with HSPICE [12]. To do that a sine wave with 2V peak-to-peak magnitude and 1KHz frequency was applied to both behavioural and transistor level circuits. The simulators were allowed to run for 5 ms with 10 (s iteration steps. Table II shows the CPU time spent for each case with the different approaches. As can be seen from the table there is an average of a 4.4 times speed-up for type I and type II cases. The speed-up for the type III faults between the behavioural and the transistor level simulations is 373.7 times. The reason why the behavioural model is so much faster than the transistor level is that type III faults are open drain faults where HSPICE needs more time to simulate that type of faults. Finally the speed-up for type IV faults is around 2.5 times. The behavioural model for type IV faults is relatively slower compared to other behavioural models in [3] due to if-then construct required in the procedural statement (Figure 17) to model the type IV faults.

Table II. Comparison of CPU times for transistor level transient HSPICE simulations against VHDL-AMS behavioural level hAMSter simulations.

	Fault Type
	The CPU time (s)

	
	Simulator

	
	hAMSter
	HSPICE

	Fault I
	90m
	400m

	Fault II
	90m
	360m

	Fault III
	100m
	37.37

	Fault IV
	140m
	350m


4 . Conclusion


Analogue fault simulation is a key requirement for analogue/mixed-signal test generation. Currently such fault simulation is limited due to the speed of analogue simulation and the large number of faults to be simulated. Simulation can be speeded up by using number of techniques. Behavioural modelling is one of those techniques. We have shown in this paper how one can increase analogue fault simulation process by using behavioural models. VHDL-AMS and hAMSter were used in this study for behavioural modelling and simulation, respectively. The advantage of the VHDL-AMS over other proprietary analogue HDLs is that the VHDL-AMS is an IEEE standard. As VHDL-AMS was standardised in 1999, it is still in its infancy and there are not many VHDL-AMS simulators currently available. It is clear that as VHDL-AMS simulators become more powerful it will be easier to model analogue/mixed-signal circuits at a higher level so as to speed-up simulation in general and analogue fault simulation in particular.

